NAG Library Routine Document

F07AJF (DGETRI)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F07AJF (DGETRI) computes the inverse of a real matrix A, where A has been factorized by F07ADF (DGETRF).

2 Specification

```
SUBROUTINE F07AJF (N, A, LDA, IPIV, WORK, LWORK, INFO)

INTEGER            N, LDA, IPIV(*), LWORK, INFO
REAL (KIND=nag_wp) A(LDA,*), WORK(max(1,LWORK))
```

The routine may be called by its LAPACK name ***dgetri***.

3 Description

F07AJF (DGETRI) is used to compute the inverse of a real matrix A, the routine must be preceded by a call to F07ADF (DGETRF), which computes the LU factorization of A as $A = PLU$. The inverse of A is computed by forming U^{-1} and then solving the equation $XPL = U^{-1}$ for X.

4 References

Du Croz J J and Higham N J (1992) Stability of methods for matrix inversion *IMA J. Numer. Anal.* **12** 1–19

5 Parameters

1: N – INTEGER *Input*

On entry: n, the order of the matrix A.

Constraint: $\mathrm{N} \geq 0$.

2: A(LDA,$*$) – REAL (KIND=nag_wp) array *Input/Output*

Note: the second dimension of the array A must be at least $\max(1, \mathrm{N})$.

On entry: the LU factorization of A, as returned by F07ADF (DGETRF).

On exit: the factorization is overwritten by the n by n matrix A^{-1}.

3: LDA – INTEGER *Input*

On entry: the first dimension of the array A as declared in the (sub)program from which F07AJF (DGETRI) is called.

Constraint: $\mathrm{LDA} \geq \max(1, \mathrm{N})$.

4: IPIV($*$) – INTEGER array *Input*

Note: the dimension of the array IPIV must be at least $\max(1, \mathrm{N})$.

On entry: the pivot indices, as returned by F07ADF (DGETRF).

5: WORK(max(1, LWORK)) – REAL (KIND=nag_wp) array *Workspace*

On exit: if INFO = 0, WORK(1) contains the minimum value of LWORK required for optimum performance.

6: LWORK – INTEGER *Input*

On entry: the dimension of the array WORK as declared in the (sub)program from which F07AJF (DGETRI) is called, unless LWORK = −1, in which case a workspace query is assumed and the routine only calculates the optimal dimension of WORK (using the formula given below).

Suggested value: for optimum performance LWORK should be at least N × *nb*, where *nb* is the ***block size***.

Constraint: LWORK ≥ max(1, N) or LWORK = −1.

7: INFO – INTEGER *Output*

On exit: INFO = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

Errors or warnings detected by the routine:

INFO < 0

If INFO = $-i$, the ith parameter had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO > 0

If INFO = i, the ith diagonal element of the factor U is zero, U is singular, and the inverse of A cannot be computed.

7 Accuracy

The computed inverse X satisfies a bound of the form:

$$|XA - I| \leq c(n)\epsilon|X|P|L||U|,$$

where $c(n)$ is a modest linear function of n, and ϵ is the ***machine precision***.

Note that a similar bound for $|AX - I|$ cannot be guaranteed, although it is almost always satisfied. See Du Croz and Higham (1992).

8 Further Comments

The total number of floating point operations is approximately $\frac{4}{3}n^3$.

The complex analogue of this routine is F07AWF (ZGETRI).

9 Example

This example computes the inverse of the matrix A, where

$$A = \begin{pmatrix} 1.80 & 2.88 & 2.05 & -0.89 \\ 5.25 & -2.95 & -0.95 & -3.80 \\ 1.58 & -2.69 & -2.90 & -1.04 \\ -1.11 & -0.66 & -0.59 & 0.80 \end{pmatrix}.$$

Here A is nonsymmetric and must first be factorized by F07ADF (DGETRF).

9.1 Program Text

```
    Program f07ajfe

!     F07AJF Example Program Text

!     Mark 24 Release. NAG Copyright 2012.

!     .. Use Statements ..
      Use nag_library, Only: dgetrf, dgetri, nag_wp, x04caf
!     .. Implicit None Statement ..
      Implicit None
!     .. Parameters ..
      Integer, Parameter               :: nin = 5, nout = 6
!     .. Local Scalars ..
      Integer                          :: i, ifail, info, lda, lwork, n
!     .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable  :: a(:,:), work(:)
      Integer, Allocatable             :: ipiv(:)
!     .. Executable Statements ..
      Write (nout,*) 'F07AJF Example Program Results'
!     Skip heading in data file
      Read (nin,*)
      Read (nin,*) n
      lda = n
      lwork = 64*n
      Allocate (a(lda,n),work(lwork),ipiv(n))

!     Read A from data file

      Read (nin,*)(a(i,1:n),i=1,n)

!     Factorize A

!     The NAG name equivalent of dgetrf is f07adf
      Call dgetrf(n,n,a,lda,ipiv,info)

      Write (nout,*)
      Flush (nout)
      If (info==0) Then

!       Compute inverse of A

!       The NAG name equivalent of dgetri is f07ajf
        Call dgetri(n,a,lda,ipiv,work,lwork,info)

!       Print inverse

!       ifail: behaviour on error exit
!              =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft
        ifail = 0
        Call x04caf('General',' ',n,n,a,lda,'Inverse',ifail)

      Else
        Write (nout,*) 'The factor U is singular'
      End If

    End Program f07ajfe
```

9.2 Program Data

```
F07AJF Example Program Data
  4                          :Value of N
  1.80   2.88   2.05  -0.89
  5.25  -2.95  -0.95  -3.80
  1.58  -2.69  -2.90  -1.04
 -1.11  -0.66  -0.59   0.80   :End of matrix A
```

9.3 Program Results

```
 F07AJF Example Program Results

 Inverse
                1          2          3          4
 1       1.7720     0.5757     0.0843     4.8155
 2      -0.1175    -0.4456     0.4114    -1.7126
 3       0.1799     0.4527    -0.6676     1.4824
 4       2.4944     0.7650    -0.0360     7.6119
```